An Answer/Originate Modem

Ronald G Parsons
9001 Laurel Grove Dr
Austin TX 78758

One of the few and nearly universal methods of exchanging data between diverse microprocessors is by means of data transmission over switched telephone facilities. Most other means of data exchange such as floppy disk or cassette tape are specific to one or a few microcomputers. But data transmission over phone lines is nearly independent of the microprocessors involved and the method or speed of the mass data storage used by either processor.

To transmit data at reasonable speeds over a telephone line, a *modem* is used to convert digital signals to an analog form for transmission over the telephone network. "Modem" is a hybrid of the words modulator and demodulator. A modem must be used because the telephone network was designed for analog voice transmission and not for digital data. The telephone network has an audio bandwidth of approximately 3000 Hz, so the modem must condition the signals to fit within this bandwidth.

Since communication usually involves data transmission in both directions, a convention has been established so that two sets of data traveling in opposite directions do not interfere with each other. The Bell 103 type of modem uses designated audio frequencies for binary 0 and 1. One of the pair of communicating entities is arbitrarily designated as the originating end and the other the answering end. As the words imply, the originating end usually originates the telephone call and the answering end usually answers, but this is not necessary. All that is necessary is for one of the pair to agree to call itself the answerer and the other the originator.

The originating end transmits a binary 0 (sometimes called a space) as a tone of 1070 Hz and a binary 1 (sometimes called a mark) as a tone of 1270 Hz. The originating end also *receives* spaces and marks as tones of 2025 Hz and 2225 Hz, respectively. The answering end has the transmit and receive frequencies interchanged. The Bell 103 modem translates serial data from voltage levels to these audio tones capable of being transmitted over standard telephone lines at a data rate from 0 to 300 bps.

> The telephone network was designed for analog voice transmission, not digital data.

A data bit is usually translated first by a terminal or microcomputer to standard voltage levels defined by an Electronic Industries Association (EIA) standard known as RS-232C. This standard defines a space as a voltage level between +5 V and +15 V and a mark as a voltage level between −5 V and −15 V. Voltages between −5 V and +5 V have undefined meaning. These signals are capable of being transmitted over wire cable for distances of several hundred feet at speeds up to several thousand bits per second.

The modem described in this article uses RS-232C levels between the processor or terminal and the telephone line; it connects to the telephone line through a device called a data access arrangement (DAA). This device has two common types: the CBS data coupler, which uses RS-232C levels to interface with the modem; and the simpler CBT data coupler, which uses contact closures (ie: switches or relays) for the modem interface. The CBT type is used in this design for simplicity. Motorola's *Application Note AN-747* entitled "Low-Speed-Modem System Design using the MC6860" discusses the interface to either coupler.

The most complicated and troublesome parts of a modem are usually the filters used to separate and purify the transmitted and received audio tones. It is not uncommon for filters for the transmit and receive frequencies each to contain several operational amplifiers and many precision resistors and capacitors. The filters used in this design, however, are available as "miniModem" building blocks from Cermetek Microelectronics, 660 National Ave, Mountain View CA 94043. They require no adjustments and few external components.

Two filters are used. One, the CH1262, is a switchable, dual-channel, transmit filter and line hybrid. The center frequency of the filter is

chosen to be 1170 Hz or 2125 Hz by changing the DC voltage on the channel-select pins. The other, the CH1267, is a switchable, dual-channel, receive filter and limiter. It is necessary for us to be able to switch the center frequencies of the filters so the modem can be used as an originate or an answer modem.

The functions of modulation, demodulation, and control are performed by a Motorola MC6860 metal-oxide semiconductor/large-scale integration (MOS/LSI) modem chip. After conversion to transistor-transistor logic (TTL) levels, the modulator section of the 6860 converts serial digital data into analog frequencies. It does this by digitally synthesizing a sine wave at one of the space and mark frequencies. This signal is filtered and amplified by the transmit filter. The demodulator section of the 6860 detects the presence of a mark or space frequency and presents a digital 0 or 1 output to the terminal or computer. The receive-signal input to the 6860 must be a 50% duty-cycle, TTL signal that is filtered and limited (ie: amplified and clipped).

Several supervisory control functions are provided by the 6860. The 6860 places the modem into answer mode (if a ring indication is detected) or into originate mode (if a handset-off-hook condition is detected). If the data terminal is ready, the detection of the ring creates an answer phone signal to the DAA. A mode-signal output from the 6860 is used to control the switchable filters to ensure that the correct set of signal pairs are used. A clear-to-send (CTS) signal is also created to indicate to the terminal or computer the establishment of a communication link.

Constructing the Modem

Figure 1 shows the schematic diagram for the modem. The signals from the terminal or computer to and from the modem are first converted from RS-232C levels to TTL levels by the 1488 and 1489A integrated circuits. The request-to-send (RTS) signal is not used by the 6860, but is used by the support circuitry to control pulse dialing and setting the answer/originate mode. The 1458 dual operational amplifier is used to convert the TTL-level mode signal, as possibly modified by the test/normal switch, to a +12 V or −12 V signal sent to switch the filters between originate and answer. The 301A operational amplifier is used to limit the received signal. The 3.9 V zener diode causes the output of the operational amplifier to be TTL compatible and the TTL gate helps square up the limited signal. The 200 k-ohm variable resistor on the CH1262 is used to set the transmit level to 0 dBm (ie: 1 mW at 600 ohms or 0.7 V RMS).

If the modem is powered up with the ready-to-send line active (ie: at +5 V to +12 V), the modem is in originate mode and the answer-phone signal from the 6860 commands the DAA telephone interface to take the phone line off hook. The telephone may then be dialed by pulsing the ready-to-send line off and on under software control. An assembly-language program for an 8080 to do automatic dialing is shown in listing 1.

If the modem is powered up and the ready-to-send line is off (ie: −5 V to −12 V), the modem will wait for a ring indication from the DAA

Text continued on page 34

Circle 18 on inquiry card. →

Number	Type	+5 V	GND	−12 V	+12 V
IC1	1488		7	1	14
IC2	1489A	14	7		
IC3	6860	12	1		
IC4	7407	14	7		
IC5	LM301			4	7
IC6	1458			4	8
IC7	7486	14	7		
IC8	7404	14	7		
IC9	7400	14	7		
IC10	CH1262		5	13	12
IC11	CH1267		18	9	7



Figure 1: *Schematic diagram of the answer/originate modem. IC1 and IC2 convert the modem RS-232C signal to a digital transistor-transistor logic (TTL) level and back. IC3 is the Motorola 6860 modem integrated circuit. IC10 and IC11 are the transmit and receive filters, respectively, used to interface the modem and the telephone line.*

Listing 1: *DIAL routine to perform automatic dialing by the computer. This listing, which is designed to run as part of a CP/M-based 8080 or Z80 system, performs automatic dialing of a telephone number with the command DIAL <phone number>. If a modem answers, this program causes its computer to act as a "dumb" terminal for the computer connected to the answering modem.*

```
                ; Auto-dial program
                ; Syntax:  DIAL <phone-number>[:<signon-character>]
                ;    <signon-character> sent when CTS is asserted.
                ;
0005 =          BDOS    EQU     5       ;BDOS entry point
E00C =          TERM    EQU     0E00CH  ;Terminal simulation subroutine
0020 =          SCTS    EQU     32      ;serial CTS
0010 =          SRTS    EQU     16      ;serial RTS
00F8 =          SERST   EQU     0F8H    ;serial status port
                ;
0100            ORG     100H
                ;
0100 31FFCB     START:  LXI     SP,0CBFFH
0103 CDA101             CALL    OFFHOOK
0106 0E64               MVI     C,100   ;wait 2 seconds for dialtone
0108 CD5C01             CALL    DELAY
010B 0E64               MVI     C,100
010D CD5C01             CALL    DELAY
0110 218100             LXI     H,81H   ;use default buffer area
0113 7E         NEXT:   MOV     A,M     ;get digit
0114 23                 INX     H
0115 B7                 ORA     A
0116 CA3301             JZ      TERMINAL
0119 FE3A               CPI     ':'     ;signon-character?
011B CA2901             JZ      GETSIGNON
011E F5                 PUSH    PSW
011F CD5301             CALL    SOUT    ;echo number
0122 F1                 POP     PSW
0123 CD6A01             CALL    DIGIT
0126 C31301             JMP     NEXT
                ;
                GETSIGNON:
0129 7E                 MOV     A,M
012A B7                 ORA     A
012B C23001             JNZ     NOTCR
012E 3E0D               MVI     A,13    ;CR if character zero
                NOTCR:
0130 32AC01             STA     SIGNON
                ;
                TERMINAL:
0133 DBF8               IN      SERST
0135 E620               ANI     SCTS
0137 C23301             JNZ     TERMINAL        ;wait for clear-to-send
013A CD4A01             CALL    SETIO   ;set I/O parameters for serial port
013D 3AAC01             LDA     SIGNON
0140 B7                 ORA     A
0141 C45301             CNZ     SOUT
0144 CD0CE0     TRANS:  CALL    TERM
0147 C34401             JMP     TRANS
                ;
014A 3E01       SETIO:  MVI     A,1     ;set Sol/SOLOS I/O parameters serial
C806 =          IPORT:  EQU     0C806H
C807 =          OPORT:  EQU     0C807H
014C 3206C8             STA     IPORT
014F 3207C8             STA     OPORT
0152 C9                 RET
```

Listing 1 continued on page 32

Listing 1 continued:

```
                ;
0153 5F         SOUT:   MOV     E,A     ;write character to console
0154 0E02               MVI     C,2
0156 E5                 PUSH    H
0157 CD0500             CALL    BDOS
015A E1                 POP     H
015B C9                 RET
                ;
015C 115403     DELAY:  LXI     D,852   ; .01 times (C) seconds
                ;Adjust DE for different clock periods
015F 0D                 DCR     C
0160 F8                 RM
0161 1B         DELA1:  DCX     D
0162 7A                 MOV     A,D
0163 B3                 ORA     E
0164 C26101             JNZ     DELA1
0167 C35C01             JMP     DELAY
                ;
016A FE2D       DIGIT:  CPI     '-'     ;Call with ASCII digit in A
016C C8                 RZ              ;skip '-'
016D FE20               CPI     ' '
016F C8                 RZ              ;skip blanks
0170 FE30               CPI     '0'
0172 DAA601             JC      DIGERR
0175 FE3A               CPI     '9'+1
0177 D2A601             JNC     DIGERR  ;not an ASCII digit
017A E60F               ANI     0FH     ;subtract ASCII Bias
017C C28101             JNZ     NOTZERO
017F C60A               ADI     10      ;zero is ten
                NOTZERO:
0181 47                 MOV     B,A
                PULSE:  ;each digit is onhook for 60 ms and offhook for 40 ms
0182 0E06               MVI     C,6
0184 CD9C01             CALL    ONHOOK
0187 CD5C01             CALL    DELAY
018A 0E04               MVI     C,4
018C CDA101             CALL    OFFHOOK
018F CD5C01             CALL    DELAY
0192 05                 DCR     B
0193 C28201             JNZ     PULSE
0196 0E64               MVI     C,100
0198 CD5C01             CALL    DELAY   ;inter-digit delay
019B C9                 RET
                ;
019C 3E00       ONHOOK: MVI     A,0     ;put line on-hook
019E D3F8               OUT     SERST
01A0 C9                 RET
                ;
                OFFHOOK:                ;take line off-hook
01A1 3E10               MVI     A,SRTS
01A3 D3F8               OUT     SERST
01A5 C9                 RET
                ;
                DIGERR:                 ;not a digit - go on-hook and reboot
01A6 CD9C01             CALL    ONHOOK
01A9 C30000             JMP     0       ;boot
                ;
01AC 00         SIGNON  DB      0       ;store for sign-on character
```



Figure 2: *Schematic diagram of the optional power supply. This regulated power supply can be eliminated if the required voltages are available from a nearby computer or terminal.*

Text continued from page 26:

telephone interface. On receipt of the ring, the 6860 will bring the answer-phone line high and begin sending the transmit carrier, which is at 2225 Hz. If the modem on the other end of the line responds with its carrier, which is at 1270 Hz, the 6860 will turn clear-to-send on about a half second later. The terminal or computer can detect this and initiate whatever procedure is necessary to communicate with the originator.

Figure 1 shows four light-emitting diodes (LEDs) that can be used by the operator to monitor the operation of the modem. The functions displayed are power-on, clear-to-send, mode (with the LED on in answer mode), and off-hook.

A power-supply schematic is shown in figure 2; it supplies +5 V, +12 V, and −12 V, regulated. These voltages may be obtained from the terminal or computer if they are available. I chose to make the modem an independent device: it was wire-wrapped on a small perforated board and enclosed in a cabinet.

Modem Software

Listing 1 shows a CP/M-based, assembly-language program for an 8080 processor to perform automatic dialing to an answer modem and to initiate communication. The CP/M syntax of the program is:

DIAL <phone number>

or

DIAL <phone number>:
<logon character>

The phone number may contain blanks and hyphens that are ignored. If an invalid character is found in the phone number, the program hangs up the telephone and rebootstraps

Text continued on page 40

Listing 2: *Remote-access computer routine. This is the software needed by the computer that is connected to the answering modem of figure 1. This routine allows its computer to be controlled by a remote terminal, with the connections made by two modems and a telephone line. This routine runs on a CP/M system.*

```
                ; Remote Access to CP/M
                ;   using a Sol and SOLOS
                ;
BFE0 =          IOCODE  EQU     0BFE0H  ;Temporary storage for I/O code
C01C =          AOUT    EQU     0C01CH  ;Write to logical output unit (A)
C022 =          AINP    EQU     0C022H  ;Read logical input unit (A)
C800 =          UIPRT   EQU     0C800H  ;User defined input routine address
C802 =          UOPRT   EQU     0C802H  ;User defined output routine address
C806 =          IPORT   EQU     0C806H  ;Standard input unit number
C807 =          OPORT   EQU     0C807H  ;Standard output unit number
00F8 =          SERST   EQU     0F8H    ;Serial status port
00D4 =          DCCMD   EQU     0D4H    ;Tarbell command port
                ;
0100            ORG     100H
                ;
0100 31FFCB     START   LXI     SP,0CBFFH
0103 3E00               MVI     A,0
0105 D3F8               OUT     SERST   ;set modem for answer - RTS off
0107 3E06               MVI     A,6     ;turn disk motor off
0109 D3D4               OUT     DCCMD
010B DBF8       NOTCTS  IN      SERST   ;CTS?
010D E620               ANI     20H     ;wait for modem to answer and get response
010F C20B01             JNZ     NOTCTS  ;no
0112 3E05               MVI     A,5     ;turn disk motor on
0114 D3D4               OUT     DCCMD
0116 CD4001             CALL    DELAY   ;wait one second
0119 3E03               MVI     A,3     ;set up SOLOS for
                                        ;  user defined I/O routines
011B 3206C8             STA     IPORT
011E 3207C8             STA     OPORT
0121 21E0BF             LXI     H,IOCODE        ;store user defined I/O addresses
0124 2202C8             SHLD    UOPRT
0127 21EBBF             LXI     H,IOCODE+XIPRT-XOPRT
012A 2200C8             SHLD    UIPRT
                ; Transfer I/O code to IOCODE
012D 21E0BF             LXI     H,IOCODE
0130 0E11               MVI     C,XEND-XOPRT
0132 114A01             LXI     D,XOPRT
                TRANLOOP:
0135 1A                 LDAX    D
0136 77                 MOV     M,A
0137 0D                 DCR     C
0138 23                 INX     H
0139 13                 INX     D
013A C23501             JNZ     TRANLOOP
013D C30000             JMP     0       ;boot
                ;
0140 =          DELAY   EQU     $
0140 110000             LXI     D,0
0143 1B         DLOP1   DCX     D
0144 7A                 MOV     A,D
0145 B3                 ORA     E
0146 C24301             JNZ     DLOP1
0149 C9                 RET
                ;
                ;Relocatable user defined I/O routines
                ;
                ;Output routine - output to serial and screen
014A 3E01       XOPRT   MVI     A,1
014C CD1CC0             CALL    AOUT    ;put on serial
014F 3E00               MVI     A,0
```

Listing 2 continued on page 38

Listing 2 continued:

```
0151 CD1CC0             CALL    AOUT    ;put on screen
0154 C9                 RET
                ;
                ;Input routine - input from serial port
0155 3E01       XIPRT   MVI     A,1
0157 CD22C0             CALL    AINP    ;get serial
015A C9                 RET
                ;
015B 00         XEND:   DB      0
                ;
```

Listing 3: *Remote-user routine. This routine allows a remote user to communicate with the operator of the host computer tied to the answering modem.*

```
                ; Write to operator
                ;Syntax:  WTO <message text>
0100                    ORG     100H
0100 210000             LXI     H,0
0103 2B         BELOOP: DCX     H
0104 7D                 MOV     A,L
0105 B4                 ORA     H
0106 D3FC               OUT     0FCH    ;sound alarm port
0108 C20301             JNZ     BELOOP
010B C9                 RET             ;return to CP/M
                ;
```

Listing 4: *Remote-user routine. This routine allows a remote user to communicate with the host computer's operator; it also allows the operator to send a reply to the remote terminal.*

```
                ; Write to operator with reply
                ;Syntax:  WTOR <message text>
                ;
0100                    ORG     100H
                ;
C019 =          SOUT    EQU     0C019H
C01C =          AOUT    EQU     0C01CH
C022 =          AINP    EQU     0C022H
                ;
0100 210000     START:  LXI     H,0
                BELLOOP:
0103 2B                 DCX     H
0104 7D                 MOV     A,L
0105 B4                 ORA     H
0106 D3FC               OUT     0FCH    ;sound alarm port
0108 C20301             JNZ     BELLOOP
                REPLOOP:
010B 3E00               MVI     A,0
010D CD22C0             CALL    AINP    ;get keyboard character
0110 CA0B01             JZ      REPLOOP
0113 FE0D               CPI     13      ;done?
0115 C8                 RZ              ;return to CP/M
0116 47                 MOV     B,A
0117 CD19C0             CALL    SOUT    ;send to standard output port
                                        ;  may be user defined port
                                        ;  such as serial and display
011A C30B01             JMP     REPLOOP
                ;
```

Text continued from page 34:

CP/M. If a colon follows the phone number, the next character is sent in ASCII form to the answering modem after the clear-to-send signal is received from the answering modem. Such a logon character is often required by timesharing services. After communication is established and any logon character is sent, the program calls a terminal-simulation subroutine (TERM) that will listen for a character which was the serial line, display it on the CP/M display, and send a character of input to the CP/M console. The serial status port and bit configuration is that of a Processor Technology Sol. The subroutine SETIO must configure CP/M to send output to the serial port and receive input from the serial port. The subroutine shown is also for the Sol.

Listing 2 shows a program that will configure the operating system to be remotely accessed. The program, after starting, will wait for the telephone to ring and the modem to answer. If the caller is an originating modem, the program will configure CP/M to use the terminal on the other end of the telephone line as the display console. All data output to the remote terminal and input to CP/M from the remote terminal is echoed to the local display.

Listings 3 and 4 show small programs that can be used by the remote user to communicate with the local operator. The programs can be used only to send a message or to send a message and get a reply from the local operator. These programs are thus named Write To Operator (WTO) and Write To Operator and Reply (WTOR).

Conclusion

Once the modem is constructed and tested, a protocol is still needed to establish two-way communications between processors. Commercial timesharing services set this protocol for their customers. Personal computer users do not have a standard file and message exchange protocol, but groups such as PCNET in the San Francisco Bay area (280 Polaris Ave, Mountain View CA 94303) are working on the problem. The PCNET protocol is based on the use of modems similar to the type described in this article. ■